// File: Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}
// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}
// File: DigiPet.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.19;

import "./ERC721.sol";

contract AccessControl {
  address payable public ceoAddress;

  constructor() {
    ceoAddress = payable(msg.sender);
  }

  modifier onlyCEO() {
    require(msg.sender == ceoAddress, '');
    _;
  }

  function setCEO(address payable _newCEO) public onlyCEO {
    require(_newCEO != address(0), '');

    ceoAddress = _newCEO;
  }
}

contract DigiPet is AccessControl, ERC721 {
  using SafeMath for uint256;

  event CreatedPetCollection(
    address indexed creator,
    uint256 indexed petCollectionId,
    uint64 indexed quantity,
    string category
  );
  event SetPrice(uint256 indexed newPrice, uint256 indexed petCollectionId);
  event StartOrStoppedSelling(
    bool indexed state,
    uint256 indexed petCollectionId
  );
  event StartOrStoppedLending(
    bool indexed state,
    uint256 indexed petCollectionId
  );
  event AddedWearables(address indexed creator, uint256 indexed tokenId);
  event RemovedWearables(address indexed creator, uint256 indexed tokenId);
  event PetCollectionMetadata(
    string indexed url,
    uint256 indexed petCollectionId
  );
  event MintedPet(
    uint256 indexed petId,
    uint256 indexed petCollectionId,
    address indexed newOwner,
    uint256 price
  );
  event PetName(string indexed name, uint256 indexed petId);
  event PetDescription(string indexed description, uint256 indexed petId);

  constructor() ERC721('DigiPet', 'DP') {}

  mapping(string => uint256) private petNameToPetId;
  mapping(uint256 => address) private petIdToOwner;
  mapping(uint256 => address) private petIdToApproved;
  mapping(address => uint256) private ownerPetCount;
  mapping(uint256 => address payable) private petCollectionIdToCreator;
  mapping(uint256 => uint64) private collectionIdToPetsRemaining;
  mapping(address => uint64) private creatorPetCollectionCount;
  mapping(uint256 => uint256) private collectionIdToTotalVolume;
  mapping(uint256 => string) private collectionIdtoMetadataUrl;

  struct Pet {
    address owner;
    string category;
    Gender gender;
    string name;
    uint48 birthday;
    string description;
    uint256 petCollectionId;
    string imageCid;
    string modelCid;
  }

  struct PetCollection {
    address creator;
    string category;
    Gender gender;
    string name;
    uint48 birthday;
    string description;
    uint256[] wearables;
    uint256 price;
    bool forSale;
    bool forLoan;
    uint64 quantity;
    string imageCid;
    string modelCid;
  }

  Pet[] private pets;
  PetCollection[] private petCollections;
  uint256 public priceCreation = 0;
  uint16 public maxQuantity = 999;
  uint256 public houseFee = 500;
  uint256 public minPetPrice = 0;
  enum Gender {
    MALE,
    FEMALE,
    NEUTRAL
  }

  function createPetCollection(
    string memory _name,
    uint64 _quantity,
    string memory _category,
    Gender _gender,
    string memory _description,
    string memory _imageCid,
    string memory _modelCid
  ) public payable {
    bytes memory nameBytes = bytes(_name);
    bytes memory categoryBytes = bytes(_category);
    bytes memory descriptionBytes = bytes(_description);
    require(msg.sender != address(0), '');
    require(nameBytes.length <= 24, 'Name more than 24 characters');
    require(msg.value >= priceCreation, 'Amount sent too low');
    require(_quantity > 0, 'Quantity must be greater than 0');
    require(_quantity <= maxQuantity, 'Quantity exceeds maximum');
    require(categoryBytes.length <= 18, 'Category > 18 characters');
    require(descriptionBytes.length <= 200, 'Description > 200 characters');

    uint48 birthday = uint48(block.timestamp);

    PetCollection memory petCollection = PetCollection({
      creator: msg.sender,
      category: _category,
      gender: _gender,
      name: _name,
      birthday: birthday,
      description: _description,
      wearables: new uint256[](0),
      price: minPetPrice,
      forSale: false,
      forLoan: true,
      quantity: _quantity,
      imageCid: _imageCid,
      modelCid: _modelCid
    });
    petCollections.push(petCollection);
    uint256 petCollectionId = petCollections.length - 1;

    collectionIdToPetsRemaining[petCollectionId] = _quantity;
    creatorPetCollectionCount[msg.sender]++;
    petCollectionIdToCreator[petCollectionId] = payable(msg.sender);

    emit CreatedPetCollection(
      msg.sender,
      petCollectionId,
      _quantity,
      _category
    );
  }

  function setPetPrice(uint256 _newPrice, uint256 _petCollectionId) public {
    require(_ownsPetCollection(msg.sender, _petCollectionId), 'Sender does not own pet collection');
    require(_newPrice >= minPetPrice, 'Price below minimum');

    petCollections[_petCollectionId].price = _newPrice;

    emit SetPrice(_newPrice, _petCollectionId);
  }

  function startOrStopSelling(bool _state, uint256 _petCollectionId) public {
    require(_ownsPetCollection(msg.sender, _petCollectionId), 'Sender does not own pet collection');

    petCollections[_petCollectionId].forSale = _state;

    emit StartOrStoppedSelling(_state, _petCollectionId);
  }

  function startOrStopLending(bool _state, uint256 _petCollectionId) public {
    require(_ownsPetCollection(msg.sender, _petCollectionId), 'Sender does not own pet collection');

    petCollections[_petCollectionId].forLoan = _state;

    emit StartOrStoppedLending(_state, _petCollectionId);
  }

  function addWearablesCollection(
    uint256 _newWearablesTokenId,
    uint256 _petCollectionId
  ) public {
    require(msg.sender != address(0), '');
    require(_ownsPetCollection(msg.sender, _petCollectionId), 'Sender does not own pet collection');

    petCollections[_petCollectionId].wearables.push(_newWearablesTokenId);

    emit AddedWearables(msg.sender, _newWearablesTokenId);
  }

  function removeWearablesCollection(
    uint256 _wearablesTokenId,
    uint256 _petCollectionId
  ) public {
    require(msg.sender != address(0), '');
    require(_ownsPetCollection(msg.sender, _petCollectionId), 'Sender does not own pet collection');

    uint256[] storage wearablesArray = petCollections[_petCollectionId]
      .wearables;

    for (uint256 i = 0; i < wearablesArray.length; i++) {
      if (wearablesArray[i] == _wearablesTokenId) {
        for (uint j = i; j < wearablesArray.length - 1; j++) {
          wearablesArray[j] = wearablesArray[j + 1];
        }
        wearablesArray.pop();
      }
    }

    emit RemovedWearables(msg.sender, _wearablesTokenId);
  }

  function setPetCollectionMetadataUrl(
    string memory _url,
    uint256 _petCollectionId
  ) public {
    require(msg.sender != address(0), '');
    require(_ownsPetCollection(msg.sender, _petCollectionId), 'Sender does not own pet collection');

    collectionIdtoMetadataUrl[_petCollectionId] = _url;

    emit PetCollectionMetadata(_url, _petCollectionId);
  }

  function mintPetFromCollection(uint256 _petCollectionId) public payable {
    uint64 petsRemaining = getNumberOfPetsRemaining(_petCollectionId);
    bool collectionForSale = petCollections[_petCollectionId].forSale;
    uint256 price = getPetPrice(_petCollectionId);
    require(msg.sender != address(0), '');
    require(msg.value >= price, 'Amount sent too low');
    require(collectionForSale == true, 'Collection not for sale');
    require(petsRemaining > 0, 'Collection sold out');

    collectionIdToTotalVolume[_petCollectionId] += msg.value;
    collectionIdToPetsRemaining[_petCollectionId]--;

    uint256 contractCut = price.mul(houseFee).div(10000);
    address payable creator = getCollectionCreator(_petCollectionId);
    creator.transfer(price.sub(contractCut));

    Pet memory pet = Pet({
      owner: msg.sender,
      category: petCollections[_petCollectionId].category,
      gender: petCollections[_petCollectionId].gender,
      name: '',
      birthday: petCollections[_petCollectionId].birthday,
      description: petCollections[_petCollectionId].description,
      petCollectionId: _petCollectionId,
      imageCid: petCollections[_petCollectionId].imageCid,
      modelCid: petCollections[_petCollectionId].modelCid
    });
    pets.push(pet);
    uint256 petId = pets.length - 1;

    _transfer(address(0), msg.sender, petId);

    emit MintedPet(petId, _petCollectionId, msg.sender, price);
  }

  function setPetName(string memory _name, uint256 _petId) public {
    bytes memory nameBytes = bytes(_name);
    require(_ownsPet(msg.sender, _petId), 'Sender does not own pet');
    require(nameBytes.length > 0, 'Name is 0 characters');
    require(nameBytes.length <= 24, 'Name more than 24 characters');
    require(petNameToPetId[_name] == 0, 'Name already registered');

    petNameToPetId[_name] = _petId;

    string memory oldName = pets[_petId].name;
    if (bytes(oldName).length != 0) {
      petNameToPetId[oldName] = 0;
    }

    pets[_petId].name = _name;

    emit PetName(_name, _petId);
  }

  function setPetDescription(
    string memory _description,
    uint256 _petId
  ) public {
    bytes memory descriptionBytes = bytes(_description);
    require(_ownsPet(msg.sender, _petId), 'Sender does not own pet');
    require(descriptionBytes.length <= 200, 'More than 200 characters');

    pets[_petId].description = _description;

    emit PetDescription(_description, _petId);
  }

  function setCreactionPrice(uint256 _newPrice) public onlyCEO {
    priceCreation = _newPrice;
  }

  function setMaxQuantity(uint16 _newQuantity) public onlyCEO {
    maxQuantity = _newQuantity;
  }

  function setHouseFee(uint256 _newFee) public onlyCEO {
    require(_newFee >= 100, 'Value must be at least 100');

    houseFee = _newFee;
  }

  function setMinPetPrice(uint256 _minPrice) public onlyCEO {
    minPetPrice = _minPrice;
  }

  function getContractBalance() public view returns (uint256) {
    uint256 balance = address(this).balance;

    return balance;
  }

  function withdrawBalance(uint256 _amount) public onlyCEO {
    require(
      _amount <= address(this).balance,
      'Amount exceeds contract balance'
    );

    if (_amount == 0) {
      _amount = address(this).balance;
    }

    ceoAddress.transfer(_amount);
  }

  function getPetCollectionData(
    uint256 _petCollectionId
  )
    public
    view
    returns (
      address _creator,
      string memory _name,
      string memory _category,
      string memory _imageCid,
      string memory _modelCid,
      bool _forSale,
      bool _forLoan,
      uint256 _price,
      uint64 _remaining
    )
  {
    uint64 petsRemaining = getNumberOfPetsRemaining(_petCollectionId);

    _creator = petCollections[_petCollectionId].creator;
    _name = petCollections[_petCollectionId].name;
    _category = petCollections[_petCollectionId].category;
    _imageCid = petCollections[_petCollectionId].imageCid;
    _modelCid = petCollections[_petCollectionId].modelCid;
    _forSale = petCollections[_petCollectionId].forSale;
    _forLoan = petCollections[_petCollectionId].forLoan;
    _price = petCollections[_petCollectionId].price;
    _remaining = petsRemaining;
  }

  function getPetCollectionDetails(
    uint256 _petCollectionId
  )
    public
    view
    returns (
      Gender _gender,
      uint48 _birthday,
      string memory _description,
      uint256[] memory _wearables,
      uint64 _quantity
    )
  {
    _gender = petCollections[_petCollectionId].gender;
    _birthday = petCollections[_petCollectionId].birthday;
    _description = petCollections[_petCollectionId].description;
    _wearables = petCollections[_petCollectionId].wearables;
    _quantity = petCollections[_petCollectionId].quantity;
  }

  function getPetData(
    uint256 _petId
  )
    public
    view
    returns (
      address _owner,
      string memory _category,
      Gender _gender,
      string memory _name,
      uint48 _birthday,
      string memory _description,
      uint256 _petCollectionId,
      string memory _imageCid,
      string memory _modelCid
    )
  {
    _owner = pets[_petId].owner;
    _category = pets[_petId].category;
    _gender = pets[_petId].gender;
    _name = pets[_petId].name;
    _birthday = pets[_petId].birthday;
    _description = pets[_petId].description;
    _petCollectionId = pets[_petId].petCollectionId;
    _imageCid = pets[_petId].imageCid;
    _modelCid = pets[_petId].modelCid;
  }

  function getAllOwners() public view returns (address[] memory) {
    uint256 total = totalSupply();
    address[] memory owners = new address[](total);

    for (uint256 i = 0; i < total; i++) {
      owners[i] = petIdToOwner[i];
    }

    return owners;
  }

  function getPetsOf(address _owner) public view returns (uint256[] memory) {
    uint256 petCount = balanceOf(_owner);

    if (petCount == 0) {
      return new uint256[](0);
    } else {
      uint256[] memory result = new uint256[](petCount);
      uint256 total = totalSupply();
      uint256 resultIndex = 0;

      for (uint256 i = 0; i < total; i++) {
        if (petIdToOwner[i] == _owner) {
          result[resultIndex] = i;
          resultIndex++;
        }
      }

      return result;
    }
  }

  function getPetCollectionsOf(
    address _creator
  ) public view returns (uint256[] memory) {
    uint64 petCollectionCount = creatorPetCollectionCount[_creator];

    if (petCollectionCount == 0) {
      return new uint256[](0);
    } else {
      uint256[] memory result = new uint256[](petCollectionCount);
      uint256 total = petCollections.length;
      uint256 resultIndex = 0;

      for (uint256 i = 0; i < total; i++) {
        if (petCollectionIdToCreator[i] == _creator) {
          result[resultIndex] = i;
          resultIndex++;
        }
      }

      return result;
    }
  }

  function getPetCollectionCount(
    address _creator
  ) public view returns (uint256) {
    uint256 _count = creatorPetCollectionCount[_creator];

    return _count;
  }

  function getNumberOfPetsRemaining(
    uint256 _petCollectionId
  ) public view returns (uint64) {
    uint64 number = collectionIdToPetsRemaining[_petCollectionId];

    return number;
  }

  function getCollectionCreator(
    uint256 _petCollectionId
  ) public view returns (address payable) {
    address payable creator = petCollectionIdToCreator[_petCollectionId];

    return creator;
  }

  function getCollectionMetadataUrl(
    uint256 _petCollectionId
  ) public view returns (string memory) {
    string memory url = collectionIdtoMetadataUrl[_petCollectionId];

    return url;
  }

  function getPetPrice(uint256 _petCollectionId) public view returns (uint256) {
    return petCollections[_petCollectionId].price;
  }

  function getCollectionTotalVolume(
    uint256 _petCollectionId
  ) public view returns (uint256) {
    return collectionIdToTotalVolume[_petCollectionId];
  }

  function getNumberOfCollections() public view returns (uint256) {
    uint256 number = petCollections.length;

    return number;
  }

  function nameToPetId(string memory _petName) public view returns (uint256) {
    return petNameToPetId[_petName];
  }

  function totalSupply() public view returns (uint256) {
    uint256 supply = pets.length;

    return supply;
  }

  function transfer(address _to, uint256 _petId) public {
    require(_to != address(0), '');
    require(_ownsPet(msg.sender, _petId), 'Sender does not own pet');

    _transfer(msg.sender, _to, _petId);
  }

  function _approved(address _to, uint256 _petId) private view returns (bool) {
    return petIdToApproved[_petId] == _to;
  }

  function _ownsPet(
    address _claimant,
    uint256 _petId
  ) private view returns (bool) {
    return petIdToOwner[_petId] == _claimant;
  }

  function _ownsPetCollection(
    address _claimant,
    uint256 _petCollectionId
  ) private view returns (bool) {
    return petCollectionIdToCreator[_petCollectionId] == _claimant;
  }

  function _transfer(
    address _from,
    address _to,
    uint256 _petId
  ) internal virtual override {
    ownerPetCount[_to]++;
    petIdToOwner[_petId] = _to;
    pets[_petId].owner = _to;

    if (_from != address(0)) {
      ownerPetCount[_from]--;
      delete petIdToApproved[_petId];
    }

    emit Transfer(_from, _to, _petId);
  }

  function balanceOf(address _owner) public view override returns (uint256) {
    uint256 balance = ownerPetCount[_owner];

    return balance;
  }

  function ownerOf(uint256 _petId) public view override returns (address) {
    address owner = petIdToOwner[_petId];

    return owner;
  }

  function safeTransferFrom(
    address _from,
    address _to,
    uint256 _petId
  ) public override {
    require(_to != address(0), '');
    require(_ownsPet(_from, _petId), 'From does not own pet');
    require(_approved(msg.sender, _petId), 'Transfer not approved');

    _transfer(_from, _to, _petId);
  }

  function transferFrom(
    address _from,
    address _to,
    uint256 _petId
  ) public override {
    require(_to != address(0), '');
    require(_ownsPet(_from, _petId), 'From does not own pet');
    require(_approved(msg.sender, _petId), 'Transfer not approved');

    _transfer(_from, _to, _petId);
  }

  function approve(address _to, uint256 _petId) public override {
    require(_ownsPet(msg.sender, _petId), 'Sender does not own pet');

    petIdToApproved[_petId] = _to;

    emit Approval(msg.sender, _to, _petId);
  }

  function tokenURI(
    uint256 _petId
  ) public view override returns (string memory) {
    uint256 collectionId = pets[_petId].petCollectionId;
    string memory uri = collectionIdtoMetadataUrl[collectionId];

    return uri;
  }
}

library SafeMath {
  function mul(uint256 a, uint256 b) internal pure returns (uint256) {
    if (a == 0) {
      return 0;
    }
    uint256 c = a * b;
    assert(c / a == b);

    return c;
  }

  function div(uint256 a, uint256 b) internal pure returns (uint256) {
    uint256 c = a / b;

    return c;
  }

  function sub(uint256 a, uint256 b) internal pure returns (uint256) {
    assert(b <= a);

    return a - b;
  }
}

// File: ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}
// File: ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./IERC721Metadata.sol";
import "./Address.sol";
import "./Context.sol";
import "./Strings.sol";
import "./ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: address zero is not a valid owner");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: invalid token ID");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        _requireMinted(tokenId);

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overridden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not token owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        _requireMinted(tokenId);

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: caller is not token owner nor approved");
        _safeTransfer(from, to, tokenId, data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || isApprovedForAll(owner, spender) || getApproved(tokenId) == spender);
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits an {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Reverts if the `tokenId` has not been minted yet.
     */
    function _requireMinted(uint256 tokenId) internal view virtual {
        require(_exists(tokenId), "ERC721: invalid token ID");
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    /// @solidity memory-safe-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}
// File: IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}
// File: IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);
}
// File: IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}
// File: IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}
// File: Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}